Calcolo regione stabilità per metodo RK

```
> A := <<0,1/3,-1/3,1>|<0,0,1,-1>|<0,0,0,1>|<0,0,0,0>>;
```

$$A := \begin{bmatrix} 0 & 0 & 0 & 0 \\ \frac{1}{3} & 0 & 0 & 0 \\ -\frac{1}{3} & 1 & 0 & 0 \\ 1 & -1 & 1 & 0 \end{bmatrix} \tag{1}$$

```
> c := <0,1/3,2/3,1> ;
```

$$c := \begin{bmatrix} 0 \\ \frac{1}{3} \\ \frac{2}{3} \\ 1 \end{bmatrix} \tag{2}$$

```
> b := <1/8|3/8|3/8|1/8> ;
```

$$b := \begin{bmatrix} \frac{1}{8} & \frac{3}{8} & \frac{3}{8} & \frac{1}{8} \end{bmatrix} \tag{3}$$

Avanzamento nel caso y' = alpha y

```
> RK := y[k+1] = y[k] + b[1]*K1 + b[2]*K2 + b[3]*K3 + b[4]*K4 ;
```

$$RK := y_{k+1} = \frac{1}{8}\,K1 + \frac{3}{8}\,K2 + \frac{3}{8}\,K3 + \frac{1}{8}\,K4 + y_k \tag{4}$$

```
> FUN := (x,y) -> alpha*y ;
```

$$FUN := (x, y) \rightarrow \alpha\, y \tag{5}$$

```
> EQ1 := K1 = h*FUN(x[k]+c[1]*h, y[k]+add(A[1,j]*K||j,j=1..4)) ;
  EQ2 := K2 = h*FUN(x[k]+c[2]*h, y[k]+add(A[2,j]*K||j,j=1..4)) ;
  EQ3 := K3 = h*FUN(x[k]+c[3]*h, y[k]+add(A[3,j]*K||j,j=1..4)) ;
  EQ4 := K4 = h*FUN(x[k]+c[4]*h, y[k]+add(A[4,j]*K||j,j=1..4)) ;
```

$$EQ1 := K1 = h\,\alpha\, y_k$$

$$EQ2 := K2 = h\,\alpha\left(y_k + \frac{1}{3}\,K1\right)$$

$$EQ3 := K3 = h\,\alpha\left(y_k - \frac{1}{3}\,K1 + K2\right)$$

$$EQ4 := K4 = h\,\alpha\left(y_k + K1 - K2 + K3\right) \tag{6}$$

```
> SOLK := solve( {EQ||(1..4)},{K||(1..4)} ) ;
```

$$SOLK := \left\{K1 = h\,\alpha\, y_k,\ K2 = \frac{1}{3}\,\alpha^2\, h^2\, y_k + h\,\alpha\, y_k,\ K3 = \frac{1}{3}\,\alpha^3\, h^3\, y_k + \frac{2}{3}\,\alpha^2\, h^2\, y_k + h\,\alpha\, y_k,\right. \tag{7}$$

$$K4 = \frac{1}{3}\alpha^3 h^3 y_k + \frac{1}{3} h^4 \alpha^4 y_k + \alpha^2 h^2 y_k + h\alpha y_k\}$$

```
> RKK := subs(SOLK,RK) ;
```

$$RKK := y_{k+1} = h\alpha y_k + \frac{1}{2}\alpha^2 h^2 y_k + \frac{1}{6}\alpha^3 h^3 y_k + \frac{1}{24} h^4 \alpha^4 y_k + y_k \qquad \textbf{(8)}$$

```
> RIC := collect(subs(h=s/alpha,rhs(RKK)-lhs(RKK)),[y[k+1],y[k]]) ;
```

$$RIC := -y_{k+1} + \left(s + \frac{1}{2}s^2 + \frac{1}{6}s^3 + \frac{1}{24}s^4 + 1\right) y_k \qquad \textbf{(9)}$$

Calcolo soluzioni della ricorrenza: polinomio caratteristico

```
> p := subs(y[k+1]=z,y[k]=1,RIC) ;
```

$$p := -z + s + \frac{1}{2}s^2 + \frac{1}{6}s^3 + \frac{1}{24}s^4 + 1 \qquad \textbf{(10)}$$

```
> SOLS := solve( p, {z} ) ;
```

$$SOLS := \left\{z = s + \frac{1}{2}s^2 + \frac{1}{6}s^3 + \frac{1}{24}s^4 + 1\right\} \qquad \textbf{(11)}$$

```
> ABSXY := abs(simplify(subs(s=x+I*y,subs(SOLS,z) ))) assuming
  x::real, y::real;
```

$$ABSXY := \qquad \textbf{(12)}$$

$$\frac{1}{24}\left(x^8 + 4x^6y^2 + 6x^4y^4 + 4x^2y^6 + y^8 + 8x^7 + 24x^5y^2 + 24x^3y^4 + 8xy^6 + 40x^6 + 72x^4y^2 + 24x^2y^4 - 8y^6 + 144x^5 + 96x^3y^2 - 48xy^4 + 384x^4 + 768x^3 + 1152x^2 + 1152x + 576\right)^{1/2}$$

```
> with(plots):
> contourplot(ABSXY,x=-3..1,y=-3..3,filledregions = true,contours=[1]
  ) ;
```



3
2
1
y
−1
−2
−3
−3
−2
−1
0
1
x